
Changelog
    


  
    
  
  [Unreleased]



  
    
  
  [2.6.2] - 2024-08-10



  
    
  
  Changed


	Updated deps and CI versions
	Fix deprecated CI badge (#427 - thanks to @kianmeng once again!)


  
    
  
  [2.6.1] - 2024-04-11



  
    
  
  Added


	Allow PEM keys to specify a password (#392 thanks to @jeremyowensboggs)


  
    
  
  Fixed


	Updated erlang JOSE to deal with CVE-2023-50966 (#418 thanks to @up2jj)


  
    
  
  [2.6.0] - 2023-01-29



  
    
  
  Changed


	Allow EdDSA keys (thanks @Tarang)
	Bump JOSE to 1.11.5
	Fix after_sign spec
	handle non binary hmac sha key error (thanks @alecostard)
	Fixes typ header claim not being over writable
	doc fixes all over
	raised Elixir version to 1.13 (JOSE raised it too so we follow)


  
    
  
  [2.5.0] - 2022-06-18



  
    
  
  Changed


	Update release workflow (thanks @dolfinus)
	Migrate GitHub Actions to erlef/setup-beam (thanks @kianmeng)
	Fix typespecs (thanks @mustardnoise)
	small refactor of CI
	raised Elixir version to 1.10


  
    
  
  [2.4.1] - 2021-10-26



  
    
  
  Changed


	fix: readme refs (thanks to @danferreira)
	Fix generate_and_sign examples in guides' documentation (thanks to @f-francine)
	Remove no_return from methods that can return (thanks to @jsmestad)
	Fix typos (thanks to @kianmeng)


  
    
  
  [2.4.0] - 2021-08-15



  
    
  
  Changed


	Major docs reviews and corrections (thanks to @kianmeng, @andreasknoepfle, Jon Forsyth, @fuelen)
	Change of CI pipeline (thanks to @dolfinus)
	Now uses Elixir 1.8 as minimum (and actually test the minimum version on CI)


  
    
  
  Fixed


	Fixed arity of Joken.Config.validate (thanks to @blagh)
	Compatibility with OTP 24.0 with JOSE update to 1.11.2


  
    
  
  [2.3.0] - 2020-09-27



  
    
  
  Changed


	(@supersimple with @bryanjos) Update CHANGELOG.md (#257)
	(@victorolinasc) chore: add public PEM only signer test
	(@victorolinasc) chore: update deps
	(@victorolinasc) Adding error handling (#277)
	(@ideaMarcos) Update common_use_cases.md (#285)
	(@victorolinasc) Clean up versions and compatibility with OTP 23 (#291)


  
    
  
  Fixed


	(@woylie) fix type specs and doc (#266)


  
    
  
  [2.2.0] - 2019-11-08



  
    
  
  Added


	(@bryanjos) Update .travis.yml to deploy to hex on tag (#232)
	(@thefuture2029) Access current_time_adapter in runtime instead of compile time (#252)
	(@victorolinasc) feat: add required claims hook (#250)


  
    
  
  Changed


	Bump benchee from 0.14.0 to 1.0.1
	Bump stream_data from 0.4.2 to 0.4.3 (#227)
	Bump ex_doc from 0.19.3 to 0.20.2 (#230)
	Bump dialyxir from 1.0.0-rc.4 to 1.0.0-rc.6
	Bump credo from 1.0.2 to 1.0.5
	Bump excoveralls from 0.10.5 to 0.11.1 (#233)
	Bump ex_doc from 0.20.2 to 0.21.1 (#240)
	Bump ex_doc from 0.21.1 to 0.21.2 (#246)
	Bump excoveralls from 0.11.1 to 0.11.2 (#243)
	Bump junit_formatter from 3.0.0 to 3.0.1 (#238)
	Bump dialyxir from 1.0.0-rc.6 to 1.0.0-rc.7 (#248)
	Bump credo from 1.0.5 to 1.1.5 (#253)
	Bump excoveralls from 0.11.2 to 0.12.0 (#254)


  
    
  
  Fixed


	(@llxff) Fix small typo in "Asymmetric cryptography signers" guide (#235)
	(@polvalente) fix: treat improper token properly (#237)
	(@chulkilee) Use short identifier from SPDX License List (#255)


  
    
  
  [2.1.0] - 2019-05-27



  
    
  
  Added


	(@tgturner) Allow custom error messages on claim validation (#221)


  
    
  
  Changed


	(@sgtpepper43) Get default signer at runtime (#212)
	(@balena) Update to JOSE 1.9 and remove Jason dependency (#216)
	(@victorolinasc) chore: deps update, docs update, removed unused application (#219)


  
    
  
  Fixed


	(@maartenvanvliet) Plural time units are deprecated >= elixir1.8 (#213)
	(@oo6) Fixed documentation (#218)
	(@popo63301) fix typo (#220)
	(@HeroicEric) Fix some typos in configuration guide (#222)


  
    
  
  [2.0.1] - 2019-02-17



  
    
  
  Changed


	Get default signer at runtime (#212) @sgtpepper43
	Update to JOSE 1.9 and remove Jason dependency (#216) @balena


  
    
  
  Fixed


	Plural time units are deprecated >= elixir1.8 (#213) @maartenvanvliet
	Fixed documentation (#218) @oo6


  
    
  
  [v2.0.0] - 2019-01-02


This is a re-write with a focus on making a clearer API surface with less ambiguity and more future proof without breaking backwards compatibility once again.
For changes on versions 1.x look on the v1.5 branch.

  
    
  
  Enhancements


	Ease of key configuration. We provide optional built-in support with Elixir's Mix.Config system. See our configuration guide for more details;
	Portable configuration using Joken.Claim;
	Encapsulate your token logic in a module with Joken.Config;
	Better error handling. We provide a lot more context in error messages;
	A good performance analysis for ensuring this hot-path in APIs won't be your bottleneck. Please see our performance documentation to check what we are talking about;
	Hooks for extending Joken functionality. All core actions in Joken have a corresponding hook for extending its functionality;
	Guides for common patterns;


  
    
  
  Backwards incompatible changes


	There is no Joken.Plug module anymore. Depending on requests we can bring that back, but we believe it is better to be on a different library;
	The API surface changed a lot but you can still use Joken with the same token pattern as versions 1.x. Please see our migrating guide.



  

    
Joken

A JSON Web Token (JWT) Library.
Please, do read our comprehensive documentation and guides:
	Changelog
	Joken Overview
	Configuration
	Signers
	Asymmetric cryptography signers
	Testing your app with Joken
	JWT Common use cases
	Migrating from Joken 1.0
	Custom header arguments


  
    
  
  Usage


Add :joken to your list of dependencies in mix.exs:
def deps do
  # .. other deps
  {:joken, "~> 2.6"},
  # Recommended JSON library
  {:jason, "~> 1.4"}
end
All set! (don't forget to take a look at our comprehensive documentation and guides!)

  
    
  
  Benchmarking


Just run the benchmark script based on the supported algorithm:
mix run benchmarks/hs_benchmark.exs
mix run benchmarks/jose_hs_benchmark.exs
mix run benchmarks/pem_rs_benchmark.exs
mix run benchmarks/rs_benchmark.exs


  
    
  
  License


Copyright (c) 2014 Bryan Joseph
Licensed under the Apache License, Version 2.0 (the "License");
you may not use this file except in compliance with the License.
You may obtain a copy of the License at http://www.apache.org/licenses/LICENSE-2.0
Unless required by applicable law or agreed to in writing, software
distributed under the License is distributed on an "AS IS" BASIS,
WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
See the License for the specific language governing permissions and
limitations under the License.

  
    
  
  Disclaimer


This library would not be possible without the work of @potatosalad (Andrew Bennet). Specifically his library erlang-jose.


  

    
Joken Overview
    

Joken is a JWT (JSON Web Token) library centered on 4 core operations:
	Generating claims: for instance, time based claims.
	Signing: using a Joken.Signer.
	Verifying: also using a Joken.Signer.
	Validating: running custom validations for received claims.

These core functions are not coupled which allows you to configure Joken for only what you need: verifying and validating incoming tokens, generating tokens for other consumers or both.
It is based upon the awesome erlang-jose. Besides having a friendlier Elixir API we include some extras:
	Simple key configuration. We provide optional built-in support with Elixir's Mix.Config system. See our configuration guide for more details;
	Portable configuration using Joken.Claim;
	Well-organized token logic by including the use Joken.Config statement in your own module;
	Built-in claim generation and validation. erlang-jose is responsible only for signing and verifying token signatures. Here we provide extra tools for claims validation and generation;
	Better error handling. We provide ExUnit like error messages for claim validation and configuration errors;
	High-quality performance analysis for ensuring this hot-path in APIs won't be your bottleneck. Please see our performance documentation to learn more;
	Good defaults. Joken comes with chosen good defaults for parsing JSON (Jason) and generating claims;
	Extensible Joken functionality with hooks. All core actions in Joken have a corresponding hook for extending its functionality. See our hooks guide.


  
    
  
  JWT algorithms


Joken supports all algorithms that JOSE supports. That includes:
	All HS, RS, ES, PS signing algorithms.
	All Edwards algorithms (Ed25519, Ed25519ph, Ed448, Ed448ph)

See jose JWS algorithm support for more information.

  
    
  
  Usage


As easy as:
A key configuration:
# config/dev.exs
config :joken, default_signer: "secret"
Optionally, a signer instance:
signer = Joken.Signer.create("HS256", "secret")
A token module:
# lib/myapp/token.ex
defmodule MyApp.Token do
  use Joken.Config
end
Then, just use your module :)
{:ok, token, claims} = MyApp.Token.generate_and_sign()

extra_claims = %{"user_id" => "some_id"}
token_with_default_plus_custom_claims = MyApp.Token.generate_and_sign!(extra_claims)

{:ok, claims} = MyApp.Token.verify_and_validate(token)

# Example with a different key than the default
claims = MyApp.Token.verify_and_validate!(token, another_key)
Or with explicit signer:
signer = Joken.Signer.create("HS256", "secret")

{:ok, token, claims} = MyApp.Token.generate_and_sign(%{}, signer)

extra_claims = %{"user_id" => "some_id"}
token_with_default_plus_custom_claims = MyApp.Token.generate_and_sign!(extra_claims, signer)

{:ok, claims} = MyApp.Token.verify_and_validate(token, signer)
The default is to use HS256 with the configured binary as the key. It will generate:
	aud: defaults to "Joken"
	iss: defaults to "Joken"
	jti: defaults to &Joken.generate_jti/0
	exp: defaults to 2hs
	nbf: defaults to current time
	iat: defaults to current time

Everything is customizable in your token module. Please, see our configuration guide for all configuration options.


  

    
Configuration
    


  
    
  
  Token configuration


One of Joken's basic concepts is a map of configuration. This map has binary keys that are the claims names and Joken.Claim instances with what to do during generation or validation.
Here is an example:
# Empty token configuration
token_config = %{}

# Let's create a Joken.Claim
iss = %Joken.Claim{
         generate: fn -> "My issuer" end,
         validate: fn claim_val, claims, context -> claim_val == "My issuer" end
      }

# Now let's add it to our token config
token_config = Map.put(token_config, "iss", iss)
This configuration map is referred to as token_config. Since creating it is cumbersome, we provide some helpers:
# Same result as the first example:
token_config = %{} |> Joken.Config.add_claim("iss", fn -> "My issuer" end, &(&1 == "My issuer"))
You need at least one of the functions (validate or generate). One example of leaving one of them empty is when you are only validating tokens. In this case you might leave generate functions empty.
With your token_config created, you can pass it to functions like: Joken.generate_claims/3 or Joken.validate/4.

  
    
  
  Signer configuration


Signer is an instance of Joken.Signer. You can create one like this:
signer = Joken.Signer.create("HS256", "my secret")
This is an explicit signer creation. You can configure a signer in mix config.exs too. Please see the docs on Joken.Signer for the accepted parameters.

  
    
  
  Module approach


In Joken 2.0 you can encapsulate all your token logic in a module with Joken.Config. You do that like this:
defmodule MyAppToken do
  use Joken.Config

  # other functions here...
end
This is the recommended approach. With this macro you get some generated functions that pass your token_config automatically to Joken's functions. It also implements the Joken.Hooks behaviour so you can override any of its callbacks. Also, by default, it will look for a signer from mix config with the default_signer key.
Let's see this in more depth below.

  
    
  
  Claims generation and validation


Let's start with an example:
defmodule MyApp.Token do
  use Joken.Config
end
With this configuration, you get:
	A key configuration named :default_signer
	Your token_config will be created by Joken.Config.default_claims/1

So, if you call MyApp.Token.generate_and_sign/2 and you have a key configured with the value :default_signer you'll get a token with:
	exp: defaults to 2 hours with validation
	iat: defaults to Joken.current_time/0
	nbf: defaults to Joken.current_time/0 with validation
	iss: defaults to "Joken" with validation
	aud: defaults to "Joken" with validation
	jti: defaults to Joken.generate_jti/0

It is important to notice that this configuration is used for claims we want to either generate dynamically (like all time based claims) or validate (like iss claim that we want to ensure is the same we use to generate our tokens).

  
    
  
  Overriding token_config/0


You can customize token generation and validation by overriding the function token_config/0 in your module. Example:
defmodule MyApp.Token do
  use Joken.Config

  def token_config do
    %{}
    |> add_claim("my_key", fn -> "My custom claim" end, &(&1 == "My custom claim"))
  end
end

{:ok, token, claims} = MyApp.Token.generate_and_sign()

{:ok, claims} = MyApp.Token.verify_and_validate(token)

claims = %{"my_key" => "My custom claim"}
Please see Joken.Config docs for more info on the generated callbacks.


  

    
Signers
    

A signer is the combination of a "key" and an algorithm. That is all we need to sign and verify tokens. In JWT's vocabulary: a JWS (JSON Web Signing) with a JWK (JSON Web Key).
For each algorithm, a specific key format is expected. HS algorithms expect an octet key (a "password" like key), RS algorithms expect an RSA key and so on.

  
    
  
  Configuration


A signer is configured using the following parameters:
	signer_alg : "HS256", "HS384" and so on.
	key_pem : a binary containing a key in PEM encoding format.
	key_openssh : a binary containing a private key in OpenSSH encoding format.
	key_map : a map with the raw parameters of the key.
	key_octet : a binary used as the password for HS algorithms only.
	key_index : the index of the key on a pem or OpenSSH key set defaults to 0.

Let's see some examples:
  # RS256 with a PEM encoded key
  [
    signer_alg: "RS256",
    key_pem: """ # You can pass a PEM encoded key... See below for all options.
    -----BEGIN RSA PRIVATE KEY-----
    MIIC...xcYw==
    -----END RSA PRIVATE KEY-----
    """
  ]

  # HS512 with an octet key
  [
    signer_alg: "HS512",
    key_octet: "a very random string"
  ]

  
    
  
  Octet keys


HS algorithms (HS256, HS384, HS512) use a simple binary for their key. You can only use key_octet with HS algorithms.
There is another octet key type (OKP -> octet key pair) for use with Edwards algorithms but normally we use OpenSSH private key encoding or a map with the octets so it is not mentioned here.

  
    
  
  All other keys


Besides HS algorithms, we have several types of keys. Each type has its own set of parameters. For example, here is a full list of RSA parameters in an RSA private key:
rsa_map_key = %{
  "d" =>
    "A2gHIUmJOzRGvklIA2S8wWayCXnF8NYAhOhu7woSwjioO3HRzvd3ptegSKDpPfABJuzhy7y08ug5ZcyFbN1hJBVY8NwNzpLSUK9wmXekrbTG9MT76NAiQTxV6fYK5DXPF4Cp0qghBt-tq0kQNKx4q9QEzLb9XonmXE2a10U8EWJIs972SFGhxKzf6aq6Ri7UDK607ngQyEhVmGxr3gDJLAGQ5wOap5NYIL2ufI5FYqH-Sby_Qk7299b-w4B0fl6u8isR8OlpwMLVnD-oqOBPH-65tE82hxPV0QbSmyzmg9hlVVinJ82YRBkbcu-XG9XXOhUqJJ7kafQrYkQx6BiFKQ",
  "dp" =>
    "Useg361ca8Aem1TToW8AfjOLAAEqkkR48UPMSS2Le9D4YFtAb_ud5CK2IevYl0R-4afXUzIoeiNRg4bOTAWmTwKKlmAp4B5GzlbPzAPhwQRCxzs5MiW0K-Nw30blBLWlJYDAnVEr3T3rqtgzXFLMhR5AHqM4VhWQK7QaxgaW7TE",
  "dq" =>
    "yueW-DmyJULJlJckFXfkivSO_X1sjQurDwDfyFLAnrvgy2EqJ-iq0gBVySMGw2CgeSQegTmuKinF4anL0wy85BK8tgxDULVOpjls4ej8ZQnJ2RVEjdxZLjKh-2yw-v6mbn7goko98nkRCBYMdDUBHNVcaY9bA8kdBWi-K6DgW2E",
  "e" => "AQAB",
  "kty" => "RSA",
  "n" =>
    "xnAUUvtW3ftv25jCB-hePVCnhROqH2PACVGoCybdtMYTl8qVABAR0d6T-BRzVhJzz0-UvBNFUQyVvKAFxtbQUZN2JgAm08UJrDQszqz5tTzodWexODdPuoCaWaWge_MZGhz5PwWd7Jc4bPAu0QzSVFpBP3CovSjv48Z2Eq0_LHXVjjX_Az-WaUh94mXFyAxFI_oCygtT-il1-japS3cXJJh0WddT3VKEBRYHmxDJd_LYE-KXQt3aTDhq0vI9sG2ivtFj0dc3w_YBdr4hlcr42ujSP3wLTPpTjituwHQhYP4j-zqu7J3FYaIxU4lkK9Y_DP27RxffFI9YDPJdwFkNJw",
  "p" =>
    "5cMQg_4MrOnHI44xEs6Jyt_22DCvw3K-GY046Ls50vIf2KlRALHI65SPKfVFo5hUuHkBuWnQV46tHJU0dlmfg4svPMm_581r59yXeI8W6G4FlsSiVyhFO3P5Q5ubVs7MNaqhvaqqPqR14cVvHSqjwX5jGuGAVuLhnOhZGbtb7_U",
  "q" =>
    "3RlGNrCRU-yV7TTikKJVJCIpe8vgLBkHQ61iuICd8AyHa4sXICgf2YBFgW8CAJOHKIp8g_Nl94VYpqWvN1YVDB7sFUlRpJL2yXvTKxDzUwtM5pf_D1O6lGEMQBRY-buhZHmPf5qG93LnsSqm5YOZGpZ6t6gHtYM9A6JOIgwsYys",
  "qi" =>
    "kG5Stetls18_1fvQx8rxhX2Ais0Xg0gLDUjpE_9TYcb-utq79HVKOQ_2PJGz09hQ_teqnhXhgGMubqaktl6UOSJr6B4JgcAY7yU-34EuSxp8uKLix9BVsF2cpiC4ADhjLKP9c7IQ7X7zfs336_Reb8fh9G_zRdwEfmqFy7m28Lg"
}
This map is in the format defined by JWK spec. Although you CAN use this format for configuring RSA keys, it is most common to use other formats like PEM (Privacy Enhanced Mail) encoded.

  
    
  
  PEM - Privacy Enhanced Mail


Please, don't mind the name... This is just history being unfair. If you are curious, take a look at Wikipedia's article on PEM here.
Joken brings a facility for setting a PEM key. Just use the config option key_pem. Paste your PEM contents there and that's it. Example:
[
signer_alg: "RS512",
key_pem: """
-----BEGIN RSA PRIVATE KEY-----

-----END RSA PRIVATE KEY-----
"""
If you are creating a signer explicitly, you need to pass the PEM in a map with the key PEM. Example:
signer = Joken.Signer.create("RS512", %{"pem" => key_pem})
Inside a PEM you can put several things. It may hold more than just a private key. For Joken, though, it might get a bit funky if you pass a PEM with several things in it. After all, we only need to read a key from it.  Joken is not a library meant to be fully compliant with the PEM standard.

  
    
  
  Private vs Public keys


Many people ask why should you use an algorithm with a private/public key pair. The beauty of it is that if you generate your token with a private key anybody with a public key can verify its integrity but they can't generate a token the same way. So, the design here is that if you need another party to verify tokens (say, a client of your server) you can send it the public key and it will validate tokens generated by your private key. The other way around is not true though.
This is the main benefit. And sure is a great one :)
So, if you only call verify functions, you don't need the private key. But, if you call sign functions, you will need the private key.
One thing that might seem confusing is that with some private keys you can SIGN and VERIFY. WTH??? Yep, some private keys contain the public key too inside of them (for example with RSA keys). So, you can sign and verify, both with the same key.

  
    
  
  Benchmarks


It is preferable to use the PEM format instead of passing a map of keys with all the values. Performance-wise it is just faster. You can run the benchmarks in the benchmarks folder.
Why is this way faster? Well, to use the key we need to parse it into the erlang expected type that is not PEM nor JWKs maps. BUT, erlang can handle PEMs natively while it can't handle JWKs.

  
    
  
  Dynamic signers


All functions that receive a key argument may be passed an instance of a Joken.Signer in its place. This is a convenience for when you need dynamic configuration such as when you are retrieving the key from an endpoint.
Example:
defmodule MyCustomAuth do
  use Joken.Config
end

# Using default signer configuration
MyCustomAuth.generate_and_sign()

# Explicit Signer instance
MyCustomAuth.generate_and_sign(%{"some" => "extra claim"}, Joken.Signer.create("HS512", "secret"))


  

    
Asymmetric cryptography signers
    

It is common to divide cryptography in three categories:
	Symmetric cryptography (also called symmetric-key algorithms):	Clear text is passed to an algorithm that uses a key for encrypting and decrypting to generate cipher text;
	Examples include: AES (Advanced Encryption Standard), Blowfish, DES (Data Encryption Standard), etc;
	In the BEAM ecosystem, it is represented by the :crypto.supports()[:ciphers].


	Asymmetric cryptography (also called public-key cryptography):	It uses a pair of keys where one is private (only known to the owner) and the other is public;
	Clear text is passed to an algorithm plus the private key and generates a cipher text;
	The cipher text can be verified by the public key, that is, we can check that it was generated by the private key.


	Hash cryptography (also called one-way functions):	Clear text is passed to a function that may or may not receive a key to generate irreversible cipher text.



In the JWA specification we have algorithms that use both symmetric and asymmetric cryptography functions. Let's see them (taken from the specification):
	"alg" Param Value	Digital Signature or MAC Algorithm	Implementation requirements
	HS256	HMAC using SHA-256	Required
	HS384	HMAC using SHA-384	Optional
	HS512	HMAC using SHA-512	Optional
	RS256	RSASSA-PKCS1-v1_5 using SHA-256	Recommended
	RS384	RSASSA-PKCS1-v1_5 using SHA-384	Optional
	RS512	RSASSA-PKCS1-v1_5 using SHA-512	Optional
	ES256	ECDSA using P-256 and SHA-256	Recommended+
	ES384	ECDSA using P-384 and SHA-384	Optional
	ES512	ECDSA using P-521 and SHA-512	Optional
	PS256	RSASSA-PSS using SHA-256 and  MGF1 with SHA-256	Optional
	PS384	RSASSA-PSS using SHA-384 and MGF1 with SHA-384	Optional
	PS512	RSASSA-PSS using SHA-512 and MGF1 with SHA-512	Optional

(removed the none algorithm we don't support**)
Besides the HSxxx family of algorithms, all others use asymmetric cryptography.

  
    
  
  Using asymmetric algorithms


What is nice about public-key cryptography is that you don't need to "share a key" between parties. The one with the private key can generate tokens and publish his public key online so that anybody can check that token was generated by the owner of the private key.
In Joken that usually means a private key can be used to call Joken.encode_and_sign and Joken.verify. If only have the public key, then you can only call Joken.verify.
There are some specifications that use this: OpenID Connect being one very famous. It specifies that authentication servers must publish their keys in JWK (JSON Web Key) format on a public URL. Here are some examples:
	Google: https://www.googleapis.com/oauth2/v3/certs
	Microsoft: https://login.microsoftonline.com/common/discovery/v2.0/keys

We have a Joken Hook that can fetch these keys, turn them into signers and verify tokens. This is JokenJwks.

  
    
  
  KEY formats


Different algorithms use different key formats. For example, RSA is based on huge prime numbers arithmetic. Its keys contain data about these prime numbers and other variables in the RSA function specification.
Elliptic curve cryptography is based on elliptic curve arithmetic. Its keys represent the function that generates infinite points on a curve specification.
Because of those differences, we usually have a container for each key type. There is a specification for how to represent these keys in JSON format. That is the JSON Web Key specification. It has a JSON format for each type of key.
We recommend reading on the appendix examples for each type of key. Here is the appendix of the JSON Web Key specification with examples for public and private RSA and EC keys.
Let's see some examples on parsing asymmetric RSA keys with Joken:

  
    
  
  RSxxx keys


This algorithm uses the RSASSA-PKCS1-v1_5 that uses SHA2 hash algorithms. The base for this algorithm is the RSA public key standard. So, to use this algorithm we need a pair of RSA keys. There are many ways to generate these keys in different environments and is outside the scope of this library. Here is one of these ways just for an example:
➜  joken git:(main) openssl genrsa -out mykey.pem 4096
Generating RSA private key, 4096 bit long modulus (2 primes)
.............++++
...........................................................................................................++++

This will generate a private RSA key in PEM (privacy enhanced mail) format on a file named mykey.pem. With this key we can sign and verify a token signature.
To use it with Joken we can call one of the Joken.Signer.create variants:
	 With the RAW PEM contents:
signer = Joken.Signer.create("RS256", %{"pem" => pem_contents})

	 With the pem file in the configuration:
use Mix.Config

config :joken,
  my_rsa_key: [
    signer_alg: "RS256",
    key_pem: """
    -----BEGIN RSA PRIVATE KEY-----
    MIICWwIBAAKBg
    ...
    -----END RSA PRIVATE KEY-----
    """
  ]

signer = Joken.Signer.parse_config(:my_rsa_key)

	 With the key in JWK format:
# example of key parameters from https://tools.ietf.org/html/rfc7517#appendix-A.1
# This is for demonstration purposes. We don't allow those parameters in the map like
# "alg", "kid", "kty". Although they are part of the JWK Set specification.

key = %{
  "alg" => "RS256",
  "e" => "AQAB",
  "kid" => "2011-04-29",
  "kty" => "RSA",
  "n" => "0vx7agoebGcQSuuPiLJXZptN9nndrQmbXEps2aiAFbWhM78LhWx4cbbfAAtVT86zwu1RK7aPFFxuhDR1L6tSoc_BJECPebWKRXjBZCiFV4n3oknjhMstn64tZ_2W-5JsGY4Hc5n9yBXArwl93lqt7_RN5w6Cf0h4QyQ5v-65YGjQR0_FDW2QvzqY368QQMicAtaSqzs8KJZgnYb9c7d0zgdAZHzu6qMQvRL5hajrn1n91CbOpbISD08qNLyrdkt-bFTWhAI4vMQFh6WeZu0fM4lFd2NcRwr3XPksINHaQ-G_xBniIqbw0Ls1jF44-csFCur-kEgU8awapJzKnqDKgw"
}

signer = Joken.Signer.create(key)

	 With the key in JWK format on the config:
use Mix.Config

config :joken,
  my_rsa_public_key: [
    signer_alg: "RS256",
    key_map: %{
      "alg" => "RS256",
      "e" => "AQAB",
      "kid" => "2011-04-29",
      "kty" => "RSA",
      "n" => "0vx7agoebGcQSuuPiLJXZptN9nndrQmbXEps2aiAFbWhM78LhWx4cbbfAAtVT86zwu1RK7aPFFxuhDR1L6tSoc_BJECPebWKRXjBZCiFV4n3oknjhMstn64tZ_2W-5JsGY4Hc5n9yBXArwl93lqt7_RN5w6Cf0h4QyQ5v-65YGjQR0_FDW2QvzqY368QQMicAtaSqzs8KJZgnYb9c7d0zgdAZHzu6qMQvRL5hajrn1n91CbOpbISD08qNLyrdkt-bFTWhAI4vMQFh6WeZu0fM4lFd2NcRwr3XPksINHaQ-G_xBniIqbw0Ls1jF44-csFCur-kEgU8awapJzKnqDKgw"
    }
  ]

signer = Joken.Signer.parse_config(:my_rsa_public_key)



  
    
  
  Other algorithms


Joken has a test suite that runs all supported algorithms. Please have a look at other algorithms from our test suite.
Also, check :erlang-jose documentation about keys: https://github.com/potatosalad/erlang-jose/blob/master/examples/KEY-GENERATION.md


  

    
Testing your app with Joken
    

One common hurdle with testing tokens is that they almost always contain time sensitive claims. How can you automate the expiration of your tokens for testing?
One possible solution to this problem is to have a different token configuration for your tests. This works but is not advisable since your production code will run a different path than your test code. Your tests might pass, but things can go wrong in production.

  
    
  
  Joken time


We have introduced an adapter for producing any time-sensitive claim values. Joken.current_time/0 looks for the implementation it will use in the configuration. This is the adapter pattern that helps you mock time if you need to.
The default implementation we use is Joken.CurrentTime.OS. It uses DateTime to fetch current time in seconds.
In our tests we have a Joken.CurrentTime.Mock that can freeze or advance time in the way that we want. Please look in our test base for one possible solution.
We don't ship Joken.CurrentTime.Mock in the library as this is only one possible way of solving this. If you already have a time mocking mechanism in your app, you can make Joken use it with:
config :joken, current_time_adapter: MyTimeMock
All it needs is to implement the function current_time/0.

  
    
  
  Behaviour


It is also worth mentioning that Joken.CurrentTime is a behaviour so you can use mocking libraries like mox.


  

    
JWT Common use cases
    

This is just a collection of common examples of working with tokens using Joken's API.

  
    
  
  JWT simplest configuration


We will start from the simplest configuration possible:
defmodule Token do
  use Joken.Config
end
With this configuration you can:
	Generate a token with aud, iss, exp, jti, iat, nbf.
	Validate a token with those claims.
	Use a default signer configuration.


  
    
  
  Custom static iss claim


defmodule Token do
  use Joken.Config

  @impl true
  def token_config do
    default_claims(skip: [:iss])
    |> add_claim("iss", fn -> "My issuer" end, &(&1 == "My issuer"))
  end
end
Since iss is a default claim, we can pass that value to default_claims directly:
defmodule Token do
  use Joken.Config

  @impl true
  def token_config do
    default_claims(iss: "My custom issuer")
  end
end

  
    
  
  Custom dynamic aud claim


In this scenario we don't want a function for generating the claim value. We will generate it according to some business context. So, we skip static generation BUT we provide the claim validation all the same.
defmodule Token do
  use Joken.Config

  @impl true
  def token_config do
    default_claims(skip: [:aud])
    |> add_claim("aud", nil, &(&1 in ["My audience", "Your audience", "Her audience"]))
  end
end

# Defined at call site
Token.generate_and_sign(%{"aud" => "My audience"})

  
    
  
  Custom validation cross claims


Sometimes you need the value of another claim to validate some other claim. For example, you need the value of the role claim to validate the audience. That is fine. The validate function can receive up to 3 arguments: 1) the claim value, 2) &1, all the claims, 3) &1, &2, context.
defmodule Token do
  use Joken.Config

  @impl true
  def token_config do
    default_claims(skip: [:aud])
    |> add_claim("aud", nil, &validate_audience/2)
  end

  defp validate_audience(value, claims) do
    case claims["role"] do
      "admin" ->
        "http://myserver.com/admin"

      "user" ->
        "http://myserver.com"
    end
  end
end

  
    
  
  Signer fetched from another server


It is common to use OpenID Connect authentication servers to federate login. In this scenario, the signer configuration is usually available in what is called a well known URL.
This URL has a JWKS configuration. This tells the world that tokens generated by these servers can be validated with these keys. Normally, the key id is a claim in the token header.
We can use Joken easily in this scenario. We can abstract all the logic of fetching the signer configuration from the URL in a Hook and configure our claims validation without generation. Let's see an example:
defmodule Token do
  use Joken.Config, default_signer: nil # no signer

  # This hook implements a before_verify callback that checks whether it has a signer configuration
  # cached. If it does not, it tries to fetch it from the jwks_url.
  add_hook(JokenJwks, jwks_url: "https://someurl.com")

  @impl true
  def token_config do
    default_claims(skip: [:aud, :iss])
    |> add_claim("roles", nil, &(&1 in ["admin", "user"]))
    |> add_claim("iss", nil, &(&1 == "some server iss"))
    |> add_claim("aud", nil, &(&1 == "some server aud"))
  end
end

# We can call this by
Token.verify_and_validate(jwt)

  
    
  
  Generate a token with the user id as subject


Another common need is to generate a token with some specific data. We can even validate that data format when we receive a token. As an example, we will validate the claim "sub" as being a valid UUID but will not generate this value in Joken. This is just an example of a dynamic claim value.
defmodule Token do
  use Joken.Config

  @impl true
  def token_config do
    default_claims()
    |> add_claim("sub", nil, &is_valid_uuid/1)
  end

  # ... implementation of UUID validation
end

# We can pass the subject as extra claims
Token.generate_and_sign(%{"sub" => "uuid"})


  

    
Migrating from Joken 1.0
    

Joken 2.0 tries to fix several issues we had with the 1.x series. Some of those issues were:
	 Initialization of the json client in JOSE
The JSON adapter needed to be set every time. This is now an application configuration.

	 Confusion between dynamic and static claim value generation
Using dynamic over static claims was confusing as this feature was thrown on after version 1.0. Now it is explicit that all claim generation must be done by providing a function that is called at token generation time. You are free to implement this token generation function to return static or dynamic values.

	 Static claims
There was another confusing feature about including static claim values. For example, the API was awkward if you wanted to pass the user id to your token generation function. Now you can pass a map of claims to be added to the token. This avoids the burden of handling all possible use cases. You can still validate any claim.

	 Debugging
The error messages were not very instructive, often requiring a deep understanding of Joken in order to debug. We've made great improvements in this area.
In order to overcome most of these issues, Joken 2.0 breaks backward compatibility in several ways. We believe it was worth it. We brought in:
	Module configuration through Joken.Config which makes it really simple to configure your claims and have it encapsulated by default;
	Hook system through Joken.Hooks to extend Joken's features with simple plug-like semantics;
	Performance analysis with a faster implementation--faster than other token libraries in the elixir community;
	Improved error messages;
	Ready for future development without breaking the API again (options in claims);
	Improved testability with mocking current time implementation;
	A Jason adapter for JOSE;
	More signer configuration options;




  
    
  
  Migrating


Joken 2 has two approaches: one similar to Joken 1.x and another one using Joken.Config. Let's talk about them separately.

  
    
  
  Keeping close to Joken 1.x style


Joken 1.x was based on configuring the Joken.Token struct and then calling sign/2 or verify/3. Joken 2.0 omits the Joken.Token struct for several reasons: the name of the module was confusing and it had some side-effects like setting the JSON module on JOSE.
We still can build a token configuration and pass it to similar functions sign and verify. The token configuration is now a simple map of claim keys that must be binaries to an instance of Joken.Claim. This struct holds the functions to operate on claims.
So, with this approach, let's compare the same configuration in both versions:
# Joken 1.x
import Joken

%Joken.Token{} # empty configuration
|> with_json_module(Poison) # no built-in Jason module
|> with_validation("some_claim", &(&1 == "some value"))
|> sign(hs256("secret")) # to change the signer for test is cumbersome
|> get_compat() # compact is not a JWT terminology in any way
# Joken 2.0
import Joken.Config # more specific

token_config =
  default_claims()
  |> add_claim("some_claim", nil, &(&1 == "some value")) # explicit no generate function

Joken.generate_and_sign()

## on your config.exs
config :joken, default_signer: "secret"

## ... or if you want to keep the explicit signer creation
Joken.generate_and_sign(token_config, nil, Joken.Signer.create("HS256", "secret"))

  
    
  
  Using the new encapsulated module configuration


The same example as above can be written differently in Joken 2. We think this is better for isolating token related logic in a single module. Here is how it could be written:
defmodule MyToken do
  use Joken.Config

  @impl true
  def token_config do
    default_claims()
    |> add_claim("some_claim", nil, &(&1 == "some value"))
  end
end

# to use it all you need is:
MyToken.generate_and_sign()
You can also add custom token logic in that module like persisting it, adding an authenticate function that receives a user and a token or something similar. You could even turn it into an authentication plug adding a call(conn, opts).
Another advantage of this approach is that you can add hooks to your processing. Check the hooks guide for more information.


  

    
Custom header arguments
    

By default, a header in a token is only meant for static information. This information is used for signature verification.
Common extra claims in the header might be the key id used to sign the contents, crypto algorithms and so on.
If you need to generate extra header claims, you can do that in one of two ways: 1. use a custom signer or 2. change the application configuration.
An example of creating a custom signer with extra header claims:
test "can set key id on signer" do
  key_id = "kid"
  signer = Signer.create("HS256", "secret", %{"kid" => key_id})
  {:ok, token, _claims} = Joken.encode_and_sign(%{}, signer)
  assert %{"kid" => ^key_id, "alg" => "HS256"} = Joken.peek_header(token)
end
Another example using the application configuration:
# config/config.exs
config :joken, signer_with_key_id: [
    signer_alg: "HS256",
    key_octet: "secret",
    jose_extra_headers: %{"kid" => "my_key_id"}
  ],

# test/sometest.exs
test "can parse with key_id" do
  {:ok, token, _claims} = Joken.encode_and_sign(%{}, Signer.parse_config(:signer_with_key_id))
  assert %{"kid" => "my_key_id", "alg" => "HS256"} = Joken.peek_header(token)
end


  

    
Joken 
    



      
Joken is a library for working with standard JSON Web Tokens.
It provides 4 basic operations:
	Verify: the act of confirming the signature of the JWT;
	Validate: processing validation logic on the set of claims;
	Claim generation: generate dynamic value at token creation time;
	Signature creation: encoding header and claims and generate a signature of their value.


  
    
  
  Architecture


The core of Joken is JOSE, a library which provides all facilities to sign and verify tokens.
Joken brings an easier Elixir API with some added functionality:
	Validating claims. JOSE does not provide validation other than signature verification.
	config.exs friendly. You can optionally define your signer configuration straight in your
config.exs.
	Portable configuration. All your token logic can be encapsulated in a module with behaviours.
	Enhanced errors. Joken strives to be as informative as it can when errors happen be it at
compilation or at validation time.
	Debug friendly. When a token fails validation, a Logger debug message will show which claim
failed validation with which value. The return value, though for security reasons, does not
contain these information.
	Performance. We have a benchmark suite for identifying where we can have a better performance.
From this analysis came: Jason adapter for JOSE and other minor tweaks.


  
    
  
  Usage


Joken has 3 basic concepts:
	Portable token claims configuration.
	Signer configuration.
	Hooks.

The portable token claims configuration is a map of binary keys to Joken.Claim structs and is used
to dynamically generate and validate tokens.
A signer is an instance of Joken.Signer that encapsulates the algorithm and the key configuration
used to sign and verify a token.
A hook is an implementation of the behaviour Joken.Hooks for easy plugging into the lifecycle of
Joken operations.
There are 2 forms of using Joken:
	 Pure data structures. You can create your token configuration and signer and use them with this
module for all 4 operations: verify, validate, generate and sign.
iex> token_config = %{} # empty config
iex> token_config = Map.put(token_config, "scope", %Joken.Claim{
...>   generate: fn -> "user" end,
...>   validate: fn val, _claims, _context -> val in ["user", "admin"] end
...> })
iex> signer = Joken.Signer.create("HS256", "my secret")
iex> {:ok, claims} = Joken.generate_claims(token_config, %{"extra"=> "claim"})
iex> {:ok, jwt, claims} = Joken.encode_and_sign(claims, signer)

	 With the encapsulated module approach using Joken.Config. See the docs for Joken.Config for
more details.
iex> defmodule MyAppToken do
...>   use Joken.Config, default_signer: :pem_rs256
...>
...>   @impl Joken.Config
...>   def token_config do
...>     default_claims()
...>     |> add_claim("role", fn -> "USER" end, &(&1 in ["ADMIN", "USER"]))
...>   end
...> end
iex> {:ok, token, _claims} = MyAppToken.generate_and_sign(%{"user_id" => "1234567890"})
iex> {:ok, _claim_map} = MyAppToken.verify_and_validate(token)

Joken.Claim 
    



      
Structure for a dynamic claim. It is used for holding functions that generate
and validate claims.

Joken.Config behaviour
    



      
Main entry point for configuring Joken. This module has two approaches:

  
    
  
  Creating a map of Joken.Claim s


If you prefer to avoid using macros, you can create your configuration manually. Joken's
configuration is just a map with keys being binaries (the claim name) and the value an
instance of Joken.Claim.

  
    
  
  Example


%{"exp" => %Joken.Claim{
  generate: fn -> Joken.Config.current_time() + (2 * 60 * 60) end,
  validate: fn val, _claims, _context -> val < Joken.Config.current_time() end
}}
Since this is cumbersome and error prone, you can use this module with a more fluent API, see:
	default_claims/1
	add_claim/4


  
    
  
  Automatically load and generate functions (recommended)


Another approach is to just use Joken.Config in a module. This will load a signer configuration
(from config.exs) and a map of Joken.Claim s.

  
    
  
  Example


defmodule MyAuth do
  use Joken.Config
end
This way, Joken.Config will implement some functions for you:
	generate_claims/1: generates dynamic claims and adds them to the passed map.
	encode_and_sign/2: takes a map of claims, encodes it to JSON and signs it.
	verify/2: check for token tampering using a signer.
	validate/2: takes a claim map and a configuration to run validations.
	generate_and_sign/2: combines generation and signing.
	verify_and_validate/2: combines verification and validation.
	token_config/0: where you customize token generation and validation.

It will also add use Joken.Hooks so you can easily hook into Joken's lifecycle.

  
    
  
  Overriding functions


All callbacks in Joken.Config and Joken.Hooks are overridable. This can be used for
customizing the token configuration. All that is needed is to override the token_config/0
function returning your map of binary keys to Joken.Claim structs. Example from the
benchmark suite:
defmodule MyCustomClaimsAuth do
  use Joken.Config

  @impl true
  def token_config do
    %{} # empty claim map
    |> add_claim("name", fn -> "John Doe" end, &(&1 == "John Doe"))
    |> add_claim("test", fn -> true end, &(&1 == true))
    |> add_claim("age", fn -> 666 end, &(&1 > 18))
    |> add_claim("simple time test", fn -> 1 end, &(Joken.current_time() > &1))
  end
end

  
    
  
  Customizing default generated claims


The default claims generation is just a bypass call to default_claims/1. If one would
like to customize it, then we need only to override the token_config function:
defmodule MyCustomDefaults do
  use Joken.Config

  def token_config, do: default_claims(default_exp: 60 * 60) # 1 hour
end

  
    
  
  Options


You can pass some options to use Joken.Config to ease on your configuration:
	:default_signer: a signer configuration key in config.exs (see Joken.Signer)

Joken.CurrentTime behaviour
    



      
Behaviour for fetching current time.

    Joken.CurrentTime.OS - Joken v2.6.2
    
    

    


  
  

    
Joken.CurrentTime.OS 
    



      
Time source for default time based claims. Can be overridden in tests.

Joken.Hooks behaviour
    



      
Behaviour for defining hooks into Joken's lifecycle.
Hooks are passed to Joken functions or added to Joken.Config through the
Joken.Config.add_hook/2 macro. They can change the execution flow of a token configuration.
There are 2 kinds of hooks: before and after.
Both of them are executed in a reduce_while call and so must always return either:
	{:halt, ...} -> when you want to abort execution (other hooks won't be called)
	{:cont, ...} -> when you want to let other hooks execute


  
    
  
  Before hooks


A before hook receives as the first parameter its options and then a tuple with the input of
the function. For example, the generate_claims function receives the token configuration plus a
map of extra claims. Therefore, a before_generate hook receives:
	the hook options or [] if none are given;
	a tuple with two elements where the first is the token configuration and the second is the extra
claims map;

The return of a before hook is always the input of the next hook. Say you want to add an extra claim
with a hook. You could do so like in this example:
defmodule EnsureExtraClaimHook do
  use Joken.Hooks

  @impl true
  def before_generate(_hook_options, {token_config, extra_claims}) do
    {:cont, {token_config, Map.put(extra_claims, "must_exist", true)}}
  end
end
You could also halt execution completely on a before hook. Just use the :halt return with an error
tuple:
defmodule StopTheWorldHook do
  use Joken.Hooks

  @impl true
  def before_generate(_hook_options, _input) do
    {:halt, {:error, :stop_the_world}}
  end
end

  
    
  
  After hooks


After hooks work similar then before hooks. The difference is that it takes and returns the result of the
operation. So, instead of receiving 2 arguments it takes three:
	the hook options or [] if none are given;
	the result tuple which might be {:error, reason} or a tuple with :ok and its parameters;
	the input to the function call.

Let's see an example with after_verify. The verify function takes as argument the token and a signer. So,
an after_verify might look like this:
defmodule CheckVerifyError do
  use Joken.Hooks
  require Logger

  @impl true
  def after_verify(_hook_options, result, input) do
    case result do
      {:error, :invalid_signature} ->
        Logger.error("Check signer!!!")
        {:halt, result}

      {:ok, _claims} ->
        {:cont, result, input}
    end
  end
end
On this example we have conditional logic for different results.

  
    
  
  Joken.Config


When you create a module that has use Joken.Config it automatically implements
this behaviour with overridable functions. You can simply override a callback
implementation directly and it will be triggered when using any of the generated
functions. Example:
defmodule HookToken do
  use Joken.Config

  @impl Joken.Hooks
  def before_generate(_options, input) do
    IO.puts("Before generating claims")
    {:cont, input}
  end
end
Now if we call HookToken.generate_claims/1 it will call our callback.
Also in Joken.Config there is an imported macro for adding hooks with options. Example:
defmodule ManyHooks do
  use Joken.Config

  add_hook(JokenJwks, jwks_url: "http://someserver.com/.well-known/certs")
end
For an implementation reference, please see the source code of Joken.Hooks.RequiredClaims

Joken.Hooks.RequiredClaims 
    



      
Hook to demand claims presence.
Adding this hook to your token configuration will allow to ensure some claims are present. It
adds an after_validate/3 implementation that checks claims presence.

  
    
  
  Example


defmodule MyToken do
  use Joken.Config

  add_hook Joken.Hooks.RequiredClaims, [:claim1, :claim2]
end
On missing claims it returns: {:error, [message: "Invalid token", missing_claims: claims]}.

Joken.Signer 
    



      
Interface between Joken and JOSE for signing and verifying tokens.
In the future we plan to keep this interface but make it pluggable for other crypto
implementations like using only standard :crypto and :public_key modules. So,
avoid depending on the inner structure of this module.

Joken.Error exception
    



      
Errors for the Joken API.

